Mat 1. 2-timersprøve den 13. maj 2017.

JE 11.5.17

Opgave 1

```
> restart:with(plots):
```

En funktion f af to reelle variable er for $(x, y) \neq (0, 0)$ givet ved

```
> f:=(x,y)->y/(x^2+y^2);
```

$$f := (x, y) \to \frac{y}{x^2 + y^2}$$

```
> f(x,y);
```

$$\frac{y}{x^2 + y^2}$$

Spørgsmål 1

I (x, y)-planen betragtes de tre punkter A= (0, 1), B = (0, -1) og C= $\left(\frac{1}{2}, \frac{1}{2}\right)$.

```
> 'f(0,1)'=f(0,1);
```

$$f(0, 1) = 1$$

```
> 'f(0,-1)'=f(0,-1);
```

$$f(0, -1) = -1$$

```
> 'f(1/2,1/2)'=f(1/2,1/2);
```

$$f\left(\frac{1}{2}, \frac{1}{2}\right) = 1$$

Da f(A) = f(C) =1, så ligger A og C på den samme niveaukurve for f nemlig på niveaukurven

$f(x, y) = 1$, $(x, y) \neq (0, 0) \Leftrightarrow \frac{y}{x^2 + y^2} = 1$, $(x, y) \neq (0, 0) \Leftrightarrow$

$x^2 + y^2 - y = 0$, $(x, y) \neq (0, 0) \Leftrightarrow x^2 + \left(y - \frac{1}{2}\right)^2 = \frac{1}{4}$, $(x, y) \neq (0, 0)$,

hvilket er cirklen med centrum (0, $\frac{1}{2}$) og radius $\frac{1}{2}$ med undtagelse af punktet (0,0).

Da f(B) = -1, så ligger B ikke på denne niveaukurve for f, men på niveaukurven

$f(x, y) = -1$, $(x, y) \neq (0, 0) \Leftrightarrow x^2 + \left(y + \frac{1}{2}\right)^2 = \frac{1}{4}$, $(x, y) \neq (0, 0)$,

hvilket er cirklen med centrum (0,-$\frac{1}{2}$) og radius $\frac{1}{2}$ med undtagelse af punktet (0,0).

Gradienten for f er givet ved

$$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = \left(\frac{-2\,xy}{(x^2 + y^2)^2}, \frac{x^2 - y^2}{(x^2 + y^2)^2}\right), \ (x, y) \neq (0, 0).$$

Spørgsmål 2

Hvis 1. koordinaten for ∇f skal være 0, skal enten x eller y være 0. Det medfører, at hvis 2. koordinaten også skal være 0, så skal både x og y være 0. Men ∇f er ikke defineret i (0,0). Da således $\nabla f(x, y) \neq (0, 0)$ for alle $(x, y) \neq (0, 0)$, så har f ingen stationære punkter.

Vi betragter den begrænsede og afsluttede punktmængde $M = \{(x, y) \mid 1 \leq x^2 + y^2 \leq 4\}$.
Dvs. M er den afsluttede cirkelring mellem de to cirkler $x^2 + y^2 = 1$ (centrum $(0, 0)$ og radius 1) og $x^2 + y^2 = 4$ (centrum $(0, 0)$ og radius 2).

```
> C1:=implicitplot(x^2+y^2<4,x=-2..2,y=-2..2,scaling=constrained,
  linestyle=1):
> C2:=implicitplot(x^2+y^2>1,x=-2..2,y=-2..2,filled=true,coloring=
  [gray,white],scaling=constrained,linestyle=1):
> R1:=implicitplot(x^2+y^2=1,x=-1..1,y=-1..1,color=black,scaling=
  constrained,linestyle=1,thickness=2):
> R2:=implicitplot(x^2+y^2=4,x=-2..2,y=-2..2,color=black,scaling=
  constrained,linestyle=1,thickness=2):
> C3:=implicitplot(x^2+y^2>4,x=-2..2,y=-2..2,scaling=constrained,
  filled=true,coloring=[white,white],linestyle=2):
> display(C1,C3,C2,R1,R2,title="Cirkelringen M");
```



Det bemærkes, at det ikke er noget krav til besvarelsen at kunne tegne denne figur af M i Maple.

Spørgsmål 3

Da M er begrænset og afsluttet og da f er kontinuert i M, så har f et globalt minimum og et globalt maksimum i M. Da f hverken har stationære punkter eller undtagelsespunkter i det indre af M, så antages disse værdier på randen af M.

Randundersøgelse:

(1):

$x^2+y^2=1 \Leftrightarrow x=\pm\sqrt{1-y^2}\ ,\ y\in[-1;1]$.

$g(y)=f\left(\pm\sqrt{1-y^2},y\right)=y,\ y\in[-1;1]$.

$g'(y)=1>0$ for alle $y\in\,]-1;1[$. Dvs g er voksende.

$g(-1)=f(0,-1)=-1$ er mindst og $g(1)=f(0,1)=1$ er størst.

(2):

$x^2+y^2=4 \Leftrightarrow x=\pm\sqrt{4-y^2}\ ,\ y\in[-2;2]$.

$h(y)=f\left(\pm\sqrt{4-y^2},y\right)=\frac{1}{4}\,y\,,y\in[-2;2]$.

$h'(y)=\frac{1}{4}>0$ for alle $y\in\,]-2;2[$. Dvs h er voksende.

$h(-2)=f(0,-2)=-\frac{1}{2}$ er mindst og $h(2)=f(0,2)=\frac{1}{2}$ er størst.

Ved numerisk sammenligning af disse undersøgelser fås, at det globale minimum er -1, som antages i punktet B og at det globale maksimum er 1, som antages i punktet A.

Det bemærkes, at da M er sammenhængende, så er værdimængden $f(M)=[-1;1]$.

Opgave 2

```
> restart:
```

Om en reel funktion $f(x,y)$ oplyses, at dens approksimerende andengradspolynomium med udviklingspunkt (0, 0) er

$$P_2(x,y)=f(0,0)+f'_x(0,0)\,x+f'_y(0,0)\,y+\frac{1}{2}\left(f''_{xx}(0,0)x^2+2\,f''_{xy}(0,0)xy+f''_{yy}(0,0)y^2\right)$$

$$=2+\frac{1}{2}x^2+y^2=2+\frac{1}{2}\left(x^2+2\,y^2\right).$$

og at dens approksimerende andengradspolynomium med udviklingspunkt ($\frac{\sqrt{6}}{3}$,0) er

$$Q_2(x,y)=f\left(\frac{\sqrt{6}}{3},0\right)+f'_x\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)+f'_y\left(\frac{\sqrt{6}}{3},0\right)y$$

$$+\frac{1}{2}\left(f''_{xx}\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)^2+2\,f''_{xy}\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)y+f''_{yy}\left(\frac{\sqrt{6}}{3},0\right)y^2\right)$$

$$=\frac{4}{3}\sqrt{e}-\sqrt{e}\left(x-\frac{\sqrt{6}}{3}\right)^2+\frac{1}{3}\sqrt{e}\,y^2=\frac{4}{3}\sqrt{e}+\frac{1}{2}\left(-2\sqrt{e}\left(x-\frac{\sqrt{6}}{3}\right)^2+\frac{2}{3}\sqrt{e}\,y^2\right).$$

Spørgsmål 1

Af det givne udtryk for $P_2(x, y)$ aflæses, at

$f(0,0)=2$, $f'_x(0,0)=0$, $f'_y(0,0)=0$, $f''_{xx}(0,0)=1$, $f''_{xy}(0,0)=0$ og $f''_{yy}(0,0)=2$.

Spørgsmål 2

Hvis f har egentligt lokalt maksimum eller egentligt lokalt minimum i et punkt, så må punktet være et stationært punkt, da f ikke har undtagelsespunkter.

Da $\nabla f(0,0) = (f'_x(0,0), f'_y(0,0)) = (0,0)$ er (0,0) et stationært punkt for f.

Hessematricen for f i punktet (0,0) er

$$\mathbf{H}(0,0) = \begin{bmatrix} f''_{xx}(0,0) & f''_{xy}(0,0) \\ f''_{xy}(0,0) & f''_{yy}(0,0) \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix},$$

som har egenværdierne 1 og 2.

Da begge egenværdier for $\mathbf{H}(0,0)$ er positive, så er $f(0,0)=2$ et egentligt lokalt minimum.

Spørgsmål 3

Af det givne udtryk for $Q_2(x, y)$ aflæses,at

$$f\left(\frac{\sqrt{6}}{3}, 0\right) = \frac{4}{3}\sqrt{e},$$

$$f'_x\left(\frac{\sqrt{6}}{3}, 0\right) = 0, \ f'_y\left(\frac{\sqrt{6}}{3}, 0\right) = 0, \ f''_{xx}\left(\frac{\sqrt{6}}{3}, 0\right) = -2\sqrt{e}, \ f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) = 0 \text{ og}$$

$$f''_{yy}\left(\frac{\sqrt{6}}{3}, 0\right) = \frac{2}{3}\sqrt{e}.$$

Da $\nabla f\left(\frac{\sqrt{6}}{3}, 0\right) = \left(f'_x\left(\frac{\sqrt{6}}{3}, 0\right), f'_y\left(\frac{\sqrt{6}}{3}, 0\right)\right) = (0,0)$ er $\left(\frac{\sqrt{6}}{3}, 0\right)$ også et stationært punkt for f.

Hessematricen for f i punktet $\left(\frac{\sqrt{6}}{3}, 0\right)$ er

$$\mathbf{H}\left(\frac{\sqrt{6}}{3}, 0\right) = \begin{bmatrix} f''_{xx}\left(\frac{\sqrt{6}}{3}, 0\right) & f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) \\ f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) & f''_{yy}\left(\frac{\sqrt{6}}{3}, 0\right) \end{bmatrix} = \begin{bmatrix} -2\sqrt{e} & 0 \\ 0 & \frac{2}{3}\sqrt{e} \end{bmatrix},$$

som har egenværdierne $-2\sqrt{e}$ og $\frac{2}{3}\sqrt{e}$.

Da de to egenværdier for $\mathbf{H}\left(\frac{\sqrt{6}}{3}, 0\right)$ har modsat fortegn, så er $f\left(\frac{\sqrt{6}}{3}, 0\right)$ hverken et egentligt lokalt minimum eller et egentligt lokalt maksimum.

Opgave 3

```
> restart:with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
```

En cylinderflade F i (x, y, z)-rummet er givet ved parameterfremstillingen

```
> r:=(u,v)-><u^2-u,u^2+u,v*u>:
> r(u,v);
```

$$\begin{bmatrix} u^2 - u \\ u^2 + u \\ v\,u \end{bmatrix}$$

hvor $u \in \left[0; \frac{\sqrt{3}}{2}\right]$ og $v \in [0; 1]$.

```
> plot3d(r(u,v),u=0..sqrt(3)/2,v=0..1,orientation=[-40,80],axes=
  normal,view=[-0.3..0,0..2,-0.01..1]);
```



Spørgsmål 1

```
> ru:=diff~(r(u,v),u);
```

$$ru := \begin{bmatrix} 2u-1 \\ 2u+1 \\ v \end{bmatrix}$$

```
> rv:=diff~(r(u,v),v);
```

$$rv := \begin{bmatrix} 0 \\ 0 \\ u \end{bmatrix}$$

Fladens normalvektor er

```
> N:=kryds(ru,rv);
```

$$N := \begin{bmatrix} (2u+1)\,u \\ -(2u-1)\,u \\ 0 \end{bmatrix}$$

Den til **r** hørende Jacobifunktion er

```
> Jacobi:=simplify(sqrt(prik(N,N)))assuming u>0;
```

$$Jacobi := \sqrt{2}\,u\,\sqrt{4u^2+1}$$

$$\mathrm{Ar}(F) = \int_F 1\,\mathrm{d}\mu = \int_0^1 \int_0^{\frac{\sqrt{3}}{2}} \mathrm{Jacobi}(u,v)\,\mathrm{d}u\,\mathrm{d}v$$

```
> Int(Int(Jacobi,u=0..sqrt(3)/2),v=0..1)=int(int(Jacobi,u=0..
  sqrt(3)/2),v=0..1);
```

$$\int_0^1 \int_0^{\frac{1}{2}\sqrt{3}} \sqrt{2}\,u\,\sqrt{4u^2+1}\,\mathrm{d}u\,\mathrm{d}v = \frac{7}{12}\sqrt{2}$$

Lad L betegne den til F hørende ledekurve i (x, y)-planen.

Spørgsmål 2

L er skæringskurven mellem fladen F og (x, y)-planen.
En parameterfremstilling for L er da

```
> s:=r(u,0);
```

$$s := \begin{bmatrix} u^2-u \\ u^2+u \\ 0 \end{bmatrix}$$

hvor $u \in \left[0, \frac{\sqrt{3}}{2}\right]$.

```
> su:=diff~(s,u);
```

$$su := \begin{bmatrix} 2u-1 \\ 2u+1 \\ 0 \end{bmatrix}$$

Den til L hørende Jacobifunktion er

```
> Jacobi:=simplify(sqrt(prik(su,su)));
```

$$Jacobi := \sqrt{8u^2+2}$$

Spørgsmål 3

$$\int_L \frac{1}{2}(y-x)\,\mathrm{d}\mu = \int_0^{\frac{\sqrt{3}}{2}} \frac{1}{2}\,(y(u)-x(u))\,\mathrm{Jacobi}(u)\,\mathrm{d}u = \int_0^{\frac{\sqrt{3}}{2}} \mathrm{u\ Jacobi}(u)\,\mathrm{d}u$$

```
> Int(u*Jacobi,u=0..sqrt(3)/2)=int(u*Jacobi,u=0..sqrt(3)/2);
```

$$\int_0^{\frac{1}{2}\sqrt{3}} u\sqrt{8u^2+2}\,\mathrm{d}u = \frac{7}{12}\sqrt{2}$$

Opgave 4

```
> restart:with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
> div:=V->VectorCalculus[Divergence](V):
> rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)end
  proc:
```

Et vektorfelt i (x, y, z)-rummet er givet ved

```
> V:=(x,y,z)-><x^2,-2*y*x,z>:
> V(x,y,z);
```

$$\begin{bmatrix} x^2 \\ -2\,y\,x \\ z \end{bmatrix}$$

I (x, z)-planen betragtes et profilområde A givet ved parameterfremstillingen

```
> s:=(u,v)-><u,0,v*sin(u)>:
> s(u,v);
```

$$\begin{bmatrix} u \\ 0 \\ v\sin(u) \end{bmatrix}$$

hvor $u \in \left[0; \frac{1}{2}\pi\right]$ og $v \in [0; 1]$.

```
> plot3d(s(u,v),u=0..Pi/2,v=0..1,axes=normal,scaling=constrained);
```



Et massivt omdrejningslegeme Ω fremkommer ved at A drejes vinklen $\frac{\pi}{4}$ omkring z-aksen mod uret set fra z-aksens positive ende.

Spørgsmål 1

Parameterfremstilling for Ω

```
> r:=(u,v,w)-><u*cos(w),u*sin(w),v*sin(u)>:
> r(u,v,w);
```

$$\begin{bmatrix} u\cos(w) \\ u\sin(w) \\ v\sin(u) \end{bmatrix}$$

hvor $u \in \left[0; \frac{1}{2}\pi\right]$, $v \in [0; 1]$ og $w \in \left[0; \frac{1}{4}\pi\right]$.

Spørgsmål 2

$\partial\Omega$ er den lukkede overflade af Ω orienteret med udadrettet enhedsnormalvektor.
Af Gauss' sætning fås da

$$\mathrm{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \mathrm{Div}\,(\mathbf{V})\, d\mu = \int_0^{\frac{\pi}{4}} \int_0^1 \int_0^{\frac{\pi}{2}} \mathrm{Div}\,(\mathbf{V})(\mathbf{r}(u, v, w))\, \mathrm{Jacobi}(u, v, w)\, du\, dv\, dw$$

```
> div(V)(x,y,z);
```

$$1$$

```
> M:=<diff~(r(u,v,w),u)|diff~(r(u,v,w),v)|diff~(r(u,v,w),w)>;
```

$$M := \begin{bmatrix} \cos(w) & 0 & -u\sin(w) \\ \sin(w) & 0 & u\cos(w) \\ v\cos(u) & \sin(u) & 0 \end{bmatrix}$$

```
> Jr:=simplify(Determinant(M));
```

$$Jr := -\sin(u)\, u$$

som er ≤ 0, da $u \in \left[0; \frac{1}{2}\pi\right]$. Den til $\mathbf{r}$ hørende Jacobifunktion er da

```
> Jacobi:=-Jr;
```

$$Jacobi := \sin(u)\, u$$

```
> integranden:=1*Jacobi;
```

$$integranden := \sin(u)\, u$$

```
> Int(Int(Int(integranden,u=0..Pi/2),v=0..1),w=0..Pi/4)=int(int
  (int(integranden,u=0..Pi/2),v=0..1),w=0..Pi/4);
```

$$\int_0^{\frac{1}{4}\pi} \int_0^1 \int_0^{\frac{1}{2}\pi} \sin(u)\, u\, du\, dv\, dw = \frac{1}{4}\pi$$

Spørgsmål 3

Lad G betegne den del af overfladen af Ω, som afgrænser Ω opadtil. G er altså den omdrejningsflade, der fremkommer ved at den øvre randkurve af A i (x, z)-planen drejes vinklen $\frac{\pi}{4}$ omkring z-aksen mod uret set fra z-aksens positive ende.
En parameterfremstilling for G er da

```
> R:=r(u,1,w);
```

$$R := \begin{bmatrix} u\cos(w) \\ u\sin(w) \\ \sin(u) \end{bmatrix}$$

hvor $u \in \left[0; \frac{1}{2}\pi\right]$ og $w \in \left[0; \frac{1}{4}\pi\right]$.

```
> Ru:=diff~(R,u);
```

$$Ru := \begin{bmatrix} \cos(w) \\ \sin(w) \\ \cos(u) \end{bmatrix}$$

```
> Rw:=diff~(R,w);
```

$$Rw := \begin{bmatrix} -u\sin(w) \\ u\cos(w) \\ 0 \end{bmatrix}$$

Fladens normalvektor er

```
> N:=simplify(kryds(Ru,Rw));
```

$$N := \begin{bmatrix} -\cos(u)\,u\cos(w) \\ -\cos(u)\,u\sin(w) \\ u \end{bmatrix}$$

$\mathbf{N}(u, w)$ danner en spids vinkel med z-aksen, da dens z-koordinat $z(u, w) = u > 0$ for $u \in \left]0; \frac{1}{2}\pi\right[$.

```
> flade:=plot3d(R,u=0..Pi/2,w=0..Pi/4,scaling=constrained):
> pil:=arrow(subs(u=1,w=Pi/8,R),subs(u=1,w=Pi/8,N)):
> display(flade,pil,orientation=[-100,70]);
```



Med den valgte orientering af den lukkede randkurve ∂G for G er derfor

$$\mathbf{n}_G = \frac{\mathbf{N}(u, w)}{|\mathbf{N}(u, w)|} \text{ for } u \in \left]0; \frac{1}{2}\pi\right] \text{(højrekonvention).}$$

Af Stokes' sætning fås da

$$\text{Cirk}(\mathbf{V}, \partial G) = \text{Flux}(\mathbf{Rot}(\mathbf{V}), G) = \int_G \mathbf{n}_G \cdot \mathbf{Rot}(\mathbf{V})\, d\mu = \int_0^{\frac{\pi}{4}} \int_0^{\frac{\pi}{2}} \mathbf{N}(u, w) \cdot \mathbf{Rot}(\mathbf{V})(\mathbf{R}(u,w))du\, dw$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 0 \\ 0 \\ -2y \end{bmatrix}$$

Rotationen taget på fladen

```
> Rot:=<0,0,-2*u*sin(w)>;
```

$$Rot := \begin{bmatrix} 0 \\ 0 \\ -2\,u\,\sin(w) \end{bmatrix}$$

```
> integranden:=prik(N,Rot);
```

$$integranden := -2\,u^2\,\sin(w)$$

```
> Int(Int(integranden,u=0..Pi/2),w=0..Pi/4)=int(int
  (integranden,u=0..Pi/2),w=0..Pi/4);
```

$$\int_0^{\frac{1}{4}\pi}\int_0^{\frac{1}{2}\pi}\left(-2\,u^2\,\sin(w)\right)\,\mathrm{d}u\,\mathrm{d}w = -\frac{1}{12}\,\pi^3 + \frac{1}{24}\,\sqrt{2}\,\pi^3$$